Determinação da velocidade crítica dos eixos

* Lourenço Humberto Portela Reinaldo

Este trabalho visa determinar, usando micro-computador, linguagem basic, a velocidade crítica de um eixo, com trechos de inércias diferentes, sujeito ao peso próprio e cargas concentradas.

1. FUNDAMENTAÇÃO TEÓRICA

Sabemos que num sistema massa mola, a freqüência natural de oscilação é dada por $W = \sqrt{\frac{K}{m}} = \sqrt{\frac{g}{\delta\,(est)}}$, sendo δ (est)a elongação estática da mola.



FIG. I

No caso do sistema massa mola ser constituído por um eixo de peso desprezível, sujeito a uma carga concentrada de massa m, na expressão da sua freqüência natural de oscilação, $W = \sqrt{\frac{g}{\delta\,(est)}}$ δ (est) representa a flecha estática, como mostra a figura II.

Para a determinação desta flecha foi usado o princípio dos trabalhos virtuais. Quanto maior o número de trechos com inércia diferente, mais trabalhosa é a determinação desta flecha.



FIG. II

Imaginemos o eixo abaixo, onde a massa m tem uma excentricidade "e" em relação a este. Quando o sistema girar, com velocidade angular W, tomando a configuração da figura III, podemos escrever, usando-se o equilíbrio dinâmico: $F_c = mw^2R$.

FIG. III



$$Ky = mw^2\,(y + e) \qquad y = e\,\frac{w^2}{\frac{K}{m} - w^2}$$

Notamos que quando W aproximar-se do valor $\sqrt{\frac{K}{m}}$, a amplitude y tenderá para infinito, por menor que seja a centricidade "e" Este valor $\sqrt{\frac{K}{m}} = \sqrt{\frac{g}{\delta\,(est)}}$, é que chamaremos de velocidade crítica do eixo, e leva o sistema a ressonância, como mostra o gráfico da figura IV.



FIG. IV

Se tivermos várias cargas concentradas, a velocidade crítica do eixo poderá ser dada, aproximadamente, pela equação de DUNKERLEY:

$$\frac{1}{W^2} = \frac{1}{W_1^2} + \frac{1}{W_2^2} + \cdots + \frac{1}{W_n^2} \;\therefore\; \frac{1}{W^2} = \sum_{i=1}^{n} \frac{1}{W_i^2}$$

Sendo W_i a velocidade crítica do eixo, se atuasse apenas a carga i.

A equação acima nos permite também calcular a velocidade crítica de um eixo devido ao seu peso próprio, bastando para isto dividirmos o mesmo em n partes, e cada pedaço funcionaria como uma carga concentrada. Quanto maior o número de partes, melhor a precisão. No nosso programa, dividimos o eixo de 100 a 110 partes, dependendo do número de trechos do mesmo.

* Eng.º, Prof. Titular de Resistência dos Materiais - UNIFOR.



FIG. V

Acrescentamos que estamos determinando a primeira velocidade crítica, o harmônico fundamental, o de maior importância no funcionamento do eixo. Devemos, logicamente, evitar que este trabalhe próximo desta velocidade.

2. PROGRAMA

No programa que apresentaremos, o eixo poderá ter até 10 trechos diferentes, e funciona bi-apoiado, como é comumente usado.

O programa é auto-instrutivo, não necessitando portanto de maiores explicações para o seu uso.

Se o leitor desejar ampliar a abrangência do programa quanto ao número de trechos de inércias diferentes, basta para isto, substituir na linha 20 os valores do DIM. Se quiser alterar o número de partes de divisão do eixo, troque o valor de S na linha 30.

```
R10 CLS:PRINT"CALCULO DA VELOCIDADE
                    CRITICA DE UM EIXO"
 20 DIM L(10),D(10),N(10),C(10),
    A(10),P(10),S(10)
 30 S=100:L=0:Z=0:D=0
 40 INPUT"NUM.DE TRECHOS,DENSIDADE,MOD.
    DE ELASTICIDADE(Kg/cm2)DO EIXO:";N,G,E
 50 FOR I=1 TO N
 60 PRINT"COMPRIMENTO E DIAMETRO DO TRECHO
    ";I;"EM cm";
70 INPUT L(I),D(I)
80 S(I)=3.14*D(I)[2/4:J(I)=3.14*D(I)[4/64
90 L=L+L(I)
100 NEXT I
110 FOR I=1 TO N
120 N(I)=INT(L(I)/L*S)+1
130 C(I)=L(I)/N(I)
140 NEXT I
150 INPUT"NUMERO DE CARGAS CONCENTRADAS";NO
160 C=0:Z=0
170 IF NO=0 THEN 240
180 FOR I=1 TO NO
190 PRINT"VALOR DE P(";I;")E DISTANCIA A
     EXT.ESQ.DO EIXO";
200 INPUT P,A
210 GOSUB 370
220 Z=Z+T
230 NEXT I
240 PRINT"*****P R O C E S S A N D O******"
250 FOR I=1 TO N
260 P=G*C(I)*S(I)/1000
270 C=C+L(I)
280 FOR J=1 TO N(I)
290 A=C-(N(I)-J+0.5)*C(I)
300 GOSUB 370
310 Z=Z+T
320 NEXT J
330 NEXT I
340 W=SQR(1/Z)/2/3.14:W=INT(W*60)
350 PRINT"VELOCIDADE CRITICA:";W;"rpm"
360 END
370 D=0:B=L-A
380 M=0
390 FOR R=1 TO N
400 M=M+L(R)
410 IF A<=M THEN 430
420 NEXT R
430 M=0
440 FOR V=1 TO R
450 M=M+L(V)
460 IF V<R THEN D=D+P*B[2/(3*L[2*E*J(V)
    )*(M[3-(M-L(V))[3)
470 IF V=R THEN D=D+P*B[2/(3*L[2*E*J(V)
    )*(A[3-(M-L(V))[3)
480 NEXT V
490 M=0
500 Y=N-R+1
510 FOR V=1 TO Y
520 M=M+L(N-V+1)
530 IF V<Y THEN D=D+P*A[2/(3*L[2*E*J(N-V+1)
    )*(M[3-(M-L(N-Y+1))[3)
540 IF V=Y THEN D=D+P*A[2/(3*L[2*E*J(N-V+1)
    )*(B[3-(M-L(N-Y+1))[3)
550 NEXT V
560 T=D/981
570 RETURN
```

3. EXEMPLO



FIG. VI

```
NUM.DE TRECHOS,DENSIDADE,MOD.DE ELASTICIDADE
(Kg/cm2)DO EIXO:? 3, 7.8,2100000
COMPRIMENTO E DIAMETRO DO TRECHO 1 EM cm?
  120,8
COMPRIMENTO E DIAMETRO DO TRECHO 2 EM cm?
  150,12
COMPRIMENTO E DIAMETRO DO TRECHO 3 EM cm?
  130,10
NUMERO DE CARGAS CONCENTRADAS? 2
VALOR DE P( 1 )E DISTANCIA A EXT.ESQ.DO EIXO
  300,190
VALOR DE P( 2 )E DISTANCIA A EXT.ESQ.DO EIXO
  200,310
************ P R O C E S S A N D O *********
VELOCIDADE CRITICA: 391 rpm
READY
```

4. BIBLIOGRAFIA

SILVA Jr. J.F. *Resistência dos Materiais.* 4ed, Belo Horizonte, Edições Engenharia e Arquitetura, 1978. 456p.

HALL, A.S; HOLOWENKO, A.R. e LAUGHLIN, H.G. *Elementos Orgânicos de Máquinas.* São Paulo, Mc Graw - Hill do Brasil, 1970. 588p.